// File: contracts/TheKoreans1945Minter.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.15;

import "@openzeppelin/contracts/access/Ownable.sol";
import "./standards/IERC721G.sol";

contract TheKoreans1945Minter is Ownable {
    event SetNFTWhitelist(address nft, bool status);
    event SetUserWhitelist(address user, bool status);
    event Mint(address user, uint256 tokenId);
    event Airdrop(address[] users, uint256 initialId, uint256 lastId);

    uint256 public constant preMintTime = 1660564800; //15Aug22 12:00PM UTC
    uint256 public constant wlMintTime = 1660568400; //15Aug22 13:00PM UTC
    uint256 public constant publicMintTime = 1660654800; //16Aug22 13:00PM UTC

    IERC721G public immutable KOREANS;
    mapping(address => bool) public nftForPreMint;
    mapping(address => bool) public isListedNFT;
    mapping(address => bool) public isListedUser;
    mapping(address => mapping(address => uint256)) public mintedAmount;

    constructor(IERC721G _KOREANS, address[] memory wlNFTs, address[] memory nftsForPreMint) {
        KOREANS = _KOREANS;

        uint256 amount = wlNFTs.length;
        for (uint256 i = 0; i < amount; i++) {
            isListedNFT[wlNFTs[i]] = true;
            emit SetNFTWhitelist(wlNFTs[i], true);
        }

        amount = nftsForPreMint.length;
        require(amount == 4, "INVALID_NFTS_FOR_PREMINT");
        for (uint256 i = 0; i < amount; i++) {
            nftForPreMint[nftsForPreMint[i]] = true;
        }
    }

    function setNFTForPreMint(address[] calldata nfts, bool status) external onlyOwner {
        for (uint256 i = 0; i < nfts.length; i++) {
            nftForPreMint[nfts[i]] = status;
        }
    }

    function setNFTWhitelist(address[] calldata nfts, bool status) external onlyOwner {
        for (uint256 i = 0; i < nfts.length; i++) {
            isListedNFT[nfts[i]] = status;
            emit SetNFTWhitelist(nfts[i], status);
        }
    }

    function setUserWhitelist(address[] calldata users, bool status) external onlyOwner {
        for (uint256 i = 0; i < users.length; i++) {
            isListedUser[users[i]] = status;
            emit SetUserWhitelist(users[i], status);
        }
    }

    function airdrop(address[] calldata users) external onlyOwner {
        uint256 startId = KOREANS.totalSupply();
        for (uint256 i = 0; i < users.length; i++) {
            KOREANS.mint(users[i], startId + i);
            emit Mint(users[i], startId + i);
        }
        emit Airdrop(users, startId, startId + users.length - 1);
    }

    function mint(address nft) external {
        if (block.timestamp < preMintTime) {
            revert("NOT_STARTED_YET");
        } else if (block.timestamp < wlMintTime) {
            if (nft == address(0)) {
                require(isListedUser[msg.sender], "UNAUTHORIZED");
                delete isListedUser[msg.sender];
            } else {
                require(KOREANS.totalSupply() < 945);
                require(nftForPreMint[nft], "NOT_YET_WITH_THIS_NFT");
                uint256 _mintedAmount = mintedAmount[msg.sender][nft]++;
                require(_mintedAmount == 0, "ONLY_1_NFT_IN_PREMINT_TIME");
                require(_mintedAmount < IERC721(nft).balanceOf(msg.sender), "OUT_OF_RANGE");
            }
        } else if (block.timestamp < publicMintTime) {
            if (nft == address(0)) {
                require(isListedUser[msg.sender], "UNAUTHORIZED");
                delete isListedUser[msg.sender];
            } else {
                require(isListedNFT[nft], "INVALID_NFT_ADDRESS");
                require(mintedAmount[msg.sender][nft]++ < IERC721(nft).balanceOf(msg.sender), "OUT_OF_RANGE");
            }
        }

        uint256 tokenId = KOREANS.totalSupply();
        KOREANS.mint(msg.sender, tokenId);
        emit Mint(msg.sender, tokenId);
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: contracts/standards/IERC721G.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.15;

import "@openzeppelin/contracts/interfaces/IERC721.sol";

interface IERC721G is IERC721 {
    event SetBaseURI(string baseURI_);

    function exists(uint256 tokenId) external view returns (bool);

    function totalSupply() external view returns (uint256);

    function setBaseURI(string calldata baseURI_) external;

    function setPause(bool status) external;

    function mint(address to, uint256 tokenId) external;

    function mintBatch(address to, uint256[] calldata tokenIds) external;

    function burn(uint256 tokenId) external;

    function burnBatch(address from, uint256[] calldata tokenIds) external;

    function batchTransferFrom(
        address from,
        address to,
        uint256[] calldata tokenIds
    ) external;
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/interfaces/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (interfaces/IERC721.sol)

pragma solidity ^0.8.0;

import "../token/ERC721/IERC721.sol";

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}